

Questions and Queries

Question:

Does ages affect the likelihood that you will be involved with a crime?

Query:

```
SELECT ?Age (count( distinct ?Report) as ?Count) WHERE {  
    ?Report ns1:hasPerson ?Person .  
    ?Person ns1:hasAge ?Age .  
    FILTER (?Age > 0)  
}  
GROUP BY ?Age  
ORDER BY ?Age
```

Question:

What is the time with the least amount of crime occurred?

Query:

```
SELECT (COUNT(?report) AS ?number_of_crimes) ?hours  
WHERE {  
    {  
        SELECT (HOURS (?o) AS ?hours) ?report  
        WHERE {  
            ?report rdf:type ns1:CrimeReport .  
            ?report ns1:hasDateTime ?o  
        }  
    }  
}  
GROUP BY ?hours  
ORDER BY ?number_of_crimes  
LIMIT 5
```

Question:

What crime occurs in a given month?

Query:

```
SELECT ?crimeDescription (count(distinct ?report) as ?count)  
(MONTH(?datetime) as ?month) WHERE{  
    ?report rdf:type ns1:CrimeReport .
```

```

    ?report ns1:hasDateTime ?datetime .
    ?report ns1:hasCrime ?crime .
    ?crime ns1:hasCrimeCrimmitedDescription ?crimeDescription .
    FILTER (MONTH(?datetime) = 1) .
}
GROUP BY ?crimeDescription
ORDER BY DESC(?count)

```

Question:

"Based on your gender, how likely are you to be involved with a crime in a given neighborhood?"

Query:

```

SELECT ?area ?gender (COUNT(?person) AS ?count)
WHERE {
    {
        ?report rdf:type ns1:CrimeReport .
        ?report ns1:hasPerson ?person .
        ?person ns1:hasSex ?gender .
        ?report ns1:hasLocation ?location .
        ?location ns1:hasAreaName ?area
        FILTER (?gender!='X')
        FILTER (?gender!="")
    }
}
GROUP BY ?area ?gender
ORDER BY ?area DESC(?count)

```

Question:

What is the distribution of crimes committed against each race per month?

Query:

```

SELECT ?victim_descent ?month ?year (COUNT(DISTINCT ?report) AS
?number_of_crimes)
WHERE
{
    SELECT ?report ?victim_descent (MONTH (?o) AS ?month) (YEAR (?o)
AS ?year)

```

```

WHERE {
    ?report rdf:type ns1:CrimeReport .
    ?report ns1:hasPerson ?person .
    ?person ns1:hasDescendent ?victim_descent .
    ?report ns1:hasDateTime ?o
    FILTER (?victim_descent != "")
}
}
GROUP BY ?victim_descent ?year ?month
ORDER BY ?victim_descent ?year ?month

```

Question:

For a given url reference, what are rdf triples that contain such a reference?

Query:

Let **X** be the url reference

```

SELECT ?s ?p ?o WHERE {
    {
        SELECT (COALESCE(X) as ?s) ?p ?o
        WHERE {
            X ?p ?o
        }
    } UNION {
        SELECT ?s (COALESCE(X) as ?p) ?o
        WHERE {
            ?s X ?o
        }
    } UNION {
        SELECT ?s ?p (COALESCE(X) as ?o)
        WHERE {
            ?s ?p X
        }
    }
}

```

Question:

What is the ratio of the number of arrests of a person who committed a crime at a particular location to that of the total number of crimes located at that location?

Query:

```

SELECT (?number_of_arrests/?number_of_crimes AS ?fraction)
WHERE {
  {
    SELECT (COUNT (?arrest_report) AS ?number_of_arrests)
    WHERE {
      ?arrest_report rdf:type ns1:ArrestReport .
      ?arrest_report ns1:hasLocation ?location .
      ?location ns1:hasAddress ?a_crime_address .
      FILTER (?a_crime_address = '6TH ST')
    }
  }
  .
  {
    SELECT (COUNT (?crime_report) AS ?number_of_crimes)
    WHERE {
      ?crime_report rdf:type ns1:CrimeReport .
      ?crime_report ns1:hasLocation ?location .
      ?location ns1:hasAddress ?c_crime_address .
      FILTER (?c_crime_address = '6TH ST')
    }
  }
}

```

Question: What is the distribution of crimes across different premises?

Query:

```

SELECT ?premiseName (COUNT(?premiseName) AS ?count)
WHERE {
  ?report rdf:type ns1:CrimeReport .
  ?report ns1:hasPremise ?premise .
  ?premise ns1:hasPremiseDescription ?premiseName
}
GROUP BY ?premiseName
ORDER BY DESC(?count)

```

